

## String file name

Recently I required a program to save data to a disk file on VZ300. Unfortunately, I discovered you cannot use a string as a file name and so I developed this little program. It searches through RAM to find where the program begins and then locates the disk file handling lines and stores their

RAM location in an array. When a file is to be accessed it pokes the file-name into these locations. When the program begins, nothing will happen for a few seconds while the program searches for the required lines.

T. Hand,  
Bentleigh, Vic

```

10 GOTO 1000
20 REM LOAD FROM FILE F$
30 GOSUB 10000:REM CHANGE FILENAME
40 REM **
50 OPEN"      ",0
60 REM **
70 IN#"      ",A,B
80 REM **
90 CLOSE"      "
100 RETURN
110 :
120 REM SAVE TO FILE F$
130 GOSUB 10000:REM CHANGE FILENAME
140 REM **
150 OPEN"      ",1
160 REM **
170 PR#"      ",A,B
180 REM **
190 CLOSE"      "
200 RETURN
210 REM ERASE FILE F$
220 GOSUB 10000:REM CHANGE FILENAME
230 REM **
240 ERA"      "
250 RETURN
260 REM ^^
270 :
280 :
290 :
300 :
310 :
320 :
330 :IT IS VERY IMPORTANT TO ENTER
340 :THE LINES WITH REM **
350 :AS THESE ARE USED TO LOCATE THE
360 :PLACE TO CHANGE THE FILE NAME.
370 :
380 :THESE THREE ROUTINES ALSO SHOULD
390 :BE AT THE TOP OF THE PROGRAM
400 :TO SAVE TIME WHILE SEARCHING
410 :FOR THEIR LOCATION IN MEMORY.
420 :
430 :WHEN SAVING OR LOADING DATA,
440 :THE LINES WITH IN# AND PR#
450 :CAN BE CHANGED TO STORE YOUR
460 :OWN DATA
470 :
480 :
490 :
500 REM MAIN PROGRAM
1000 GOSUB 20000:REM INITIALIZE
1010 CLS
1020 PRINT "DO YOU WANT TO "
1030 PRINT "SAVE, RE-SAVE OR LOAD"
1040 A$=INKEY$:IF LEN(A$)=0 GOTO 1040
1050 IF A$="R" THEN GOSUB 2000
1060 IF A$="S" THEN GOSUB 3000
1070 IF A$="L" THEN GOSUB 4000
1080 GOTO 1010
1090 :
1100 :
1110 REM RE-SAVE A FILE
2000 ER=-1
2010 GOSUB 3000:REM ENTER DATA
2020 ER=0
2030 RETURN
2900 *****
2910 :THIS ROUTINE CAN BE CHANGED

```

```

2920 :TO ALLOW ENTRY OF YOUR OWN
2930 :DATA. THE ABOVE IS JUST AN
2940 :EXAMPLE OF SAVING DATA TO A
2950 :DISK FILE.
2960 *****
2980 :
2990 REM SAVE TO A FILE
3000 CLS
3010 INPUT"PLEASE ENTER THE FIRST VALUE";A
3020 INPUT"PLEASE ENTER THE SECOND VALUE";B
3030 GOSUB 5000
3040 IF ER THEN GOSUB 250
3050 GOSUB 130
3060 RETURN
3980 :
3990 REM LOAD FROM A FILE
4000 CLS
4010 GOSUB 5000
4020 GOSUB 30
4030 CLS
4040 PRINT "FIRST VALUE ENTERED WAS - "
4050 PRINT A
4060 PRINT "SECOND VALUE ENTERED WAS -"
4070 PRINT B
4080 A$=INKEY$:IF LEN(A$)<>0 GOTO 4080:REM CLEAR BUFFER
4090 PRINT:PRINT
4100 PRINT "PRESS SPACE BAR TO CONTINUE"
4110 A$=INKEY$:IF A$<>" " GOTO 4110:REM WAIT FOR SPACE
4120 RETURN
4980 :
4990 REM ASK FOR FILENAME
5000 CLS
5010 INPUT "PLEASE ENTER THE FILENAME";F$
5020 F1$=MID$(F$+"      ",1,6)
5030 RETURN
9980 :
9990 REM CHANGE FILE NAMES TO F$
10000 FOR I=1 TO 7
10010 IF F(I)=0 GOTO 10080
10020 C=0
10030 FOR J=1 TO LEN(F1$)
10040 POKE F(I)+C,ASC(MID$(F1$,J,1))
10050 C=C+1
10060 NEXT J
10070 NEXT I
10080 RETURN
19980 :
19990 REM INITIALIZE ROUTINE
20000 DIM F(7)
20010 C=1
20020 FOR I=31500 TO 33000
20030 IF NOT(PEEK(I)=42 AND PEEK(I+1)=42) GOTO 20080
20040 FOR J=I TO I+20
20050 IF PEEK(J)=34 THEN F(C)=J+1:C=C+1:GOTO 20080
20060 NEXT J
20070 PRINT "ERROR FINDING FILE NAMES":END
20080 IF PEEK(I)=94 AND PEEK(I+1)=94 GOTO 20100
20090 NEXT I
20100 RETURN

```